

## Расчет балки.

Рассмотрим изгиб в плоскости  $xu$  на примере. Расчленим балку, данную на рис.1 ниже на два элемента разрезая ее в точке А,В,С




Рис.1. Жестко-защемленная балка с сосредоточенной силой в центре балки.

Составим уравнения равновесия  $\sum Q=0$  и  $\sum M=0$  узлов А,В,С. Уравнения записаны так, чтобы можно было явно просмотреть вклад каждого элемента. Это сделано для того, чтобы легче было составить алгоритм решения

1 эл.    2 эл.

$$\begin{aligned}
 u_1 = \sum Q_A &= Q_1 + 0 + Q_R + 0 + 0 = 0 \\
 u_2 = \sum M_A &= M_1 + 0 + M_R + 0 + 0 = 0 \\
 u_3 = \sum Q_C &= Q_2 + Q_1 + 0 + 0 + -P = 0 \\
 u_4 = \sum M_C &= M_2 + M_1 + 0 + 0 + 0 = 0 \\
 u_5 = \sum Q_B &= 0 + Q_2 + 0 + Q_R + 0 = 0 \\
 u_6 = \sum M_B &= 0 + M_2 + 0 + M_R + 0 = 0
 \end{aligned} \tag{1}$$

здесь  $u_i$  - номер уравнения равновесия.

Для каждого элемента можно определить список индексов. Список индексов указывает в какие уравнения дает вклад каждый элемент. т.е.

№ эл.    Список индексов (номера уравнений)

	S(1)	S(2)	S(3)	S(4)
1	1	2	3	4
2	3	4	5	6

Для решения задачи одних только уравнений равновесия недостаточно, так как задача статически неопределима. В этом случае, составляются еще и геометрические уравнения. Для их получения, необходимы зависимости между узловыми усилиями и узловыми перемещениями. Для получения этих зависимостей, можно использовать дифференциальные уравнения изгиба

балки.

Из курса сопротивления материалов нам известно, что дифференциальное уравнение изгиба имеет вид

$$\frac{d\phi_x}{dx} = \frac{M_x}{EJ} \quad (2)$$

где  $\phi_x$  - угол поворота сечения  $x$ , а  $M_x$  - изгибающий момент в сечении  $x$ .  
 Определим выражения для  $Q_1, M_1, Q_2, M_2$




Рис.2

из рис.2 видно, что изгибающий момент в сечении  $x$  равен  $M_x = Q_1x + M_1$ ,  
 теперь решая дифференциальное уравнение (2) получим

$$\int d\phi_x = \int \frac{M_x}{EJ} dx$$

$$\int d\phi_x = \int \frac{Q_1x + M_1}{EJ} dx$$

$$\phi_x = \frac{1}{2} \frac{Q_1x^2}{EJ} + \frac{M_1x}{EJ} + C_1, \quad \text{при } x = 0, \phi_x = \phi_1, C_1 = \phi_1$$

тогда

$$\phi_x = \frac{1}{2} \frac{Q_1x^2}{EJ} + \frac{M_1x}{EJ} + \phi_1 \quad (3)$$

прогиб  $v_x$  в сечении  $x$  определим из уравнения  $\frac{dv_x}{dx} = \phi_x$ ,

$$\int dv_x = \int \phi_x dx$$

$$v_x = \frac{1}{6} \frac{Q_1x^3}{EJ} + \frac{1}{2} \frac{M_1x^2}{EJ} + \phi_1x + C_2, \quad (x = 0, C_2 = v_1) \quad (4)$$

Запишем выражения  $\phi_x$  и  $v_x$  при  $x = l$

$$\begin{aligned} \phi_2 &= \frac{1}{2} \frac{Q_1l^2}{EJ} + \frac{M_1l}{EJ} + \phi_1 \\ v_2 &= \frac{1}{6} \frac{Q_1l^3}{EJ} + \frac{1}{2} \frac{M_1l^2}{EJ} + \phi_1l + v_1 \end{aligned} \Leftrightarrow \begin{bmatrix} \frac{l^2}{2EJ} & \frac{l}{EJ} \\ \frac{l^3}{6EJ} & \frac{l^2}{2EJ} \end{bmatrix} \begin{bmatrix} Q_1 \\ M_1 \end{bmatrix} = \begin{bmatrix} -\phi_1 + \phi_2 \\ v_2 - \phi_1l - v_1 \end{bmatrix} \quad (5)$$

Решая систему получим

$$Q_1 = \frac{6EJ(l\phi_2 - 2v_2 + \phi_1 l + 2v_1)}{l^3}$$

$$M_1 = -\frac{2EJ(l\phi_2 - 3v_2 + 2\phi_1 l + 3v_1)}{l^2} \quad (6)$$

В конце балки

$$Q_2 = Q_1$$

$$M_2 = Q_1 l + M_1 \quad (7)$$

выражения для  $Q_1, M_1, Q_2, M_2$  можно записать в матричном виде

$$\begin{bmatrix} Q_1 \\ M_1 \\ Q_2 \\ M_2 \end{bmatrix} = \frac{EJ}{l^3} \begin{bmatrix} 12 & 6l & -12 & 6l \\ -6l & -4l^2 & 6l & -2l^2 \\ 12 & 6l & -12 & 6l \\ 6l & 2l^2 & -6l & 4l^2 \end{bmatrix} \begin{bmatrix} v_1 \\ \phi_1 \\ v_2 \\ \phi_2 \end{bmatrix} \quad (8)$$

Запись уравнений в матричном виде очень удобно при программировании, так как задачи по МКЭ сводятся к решению системы линейных уравнений. А алгоритмы разработаны с учетом того, что коэффициенты системы уравнений записываются в двумерные массивы

Учитывая направления  $M_1$  и  $Q_2$  на рис.2 и в расчетной схеме на рис.1 надо знаки при  $M_1$  и  $Q_2$  поменять на противоположный т.е.

$$\begin{bmatrix} Q_1 \\ M_1 \\ Q_2 \\ M_2 \end{bmatrix} = \frac{EJ}{l^3} \begin{bmatrix} 12 & 6l & -12 & 6l \\ 6l & 4l^2 & -6l & 2l^2 \\ -12 & -6l & 12 & -6l \\ 6l & 2l^2 & -6l & 4l^2 \end{bmatrix} \begin{bmatrix} v_1 \\ \phi_1 \\ v_2 \\ \phi_2 \end{bmatrix} = [K][u] \quad (9)$$

Все столбцы в уравнениях равновесия (1) можно записать в матричном виде. Например,

1 столбец

$$\begin{matrix} S_1 = 1 \\ S_2 = 2 \\ S_3 = 3 \\ S_4 = 4 \end{matrix} \begin{bmatrix} Q_1 \\ M_1 \\ Q_2 \\ M_2 \\ 0 \\ 0 \end{bmatrix} = \begin{matrix} k_{11} & k_{12} & k_{13} & k_{14} & 0 & 0 \\ k_{21} & k_{22} & k_{23} & k_{24} & 0 & 0 \\ k_{31} & k_{32} & k_{33} & k_{34} & 0 & 0 \\ k_{41} & k_{42} & k_{43} & k_{44} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{matrix} \begin{bmatrix} v_1 \\ \phi_1 \\ v_2 \\ \phi_2 \\ v_3 \\ \phi_3 \end{bmatrix} = [A]_1 [U]_1$$

$S_1 \quad S_2 \quad S_3 \quad S_4$

2 столбец

$$\begin{matrix} S_1 = 3 \\ S_2 = 4 \\ S_3 = 5 \\ S_4 = 6 \end{matrix} \begin{bmatrix} 0 \\ 0 \\ Q_1 \\ M_1 \\ Q_2 \\ M_2 \end{bmatrix} = \begin{matrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & k_{11} & k_{12} & k_{13} & k_{14} \\ 0 & 0 & k_{21} & k_{22} & k_{23} & k_{24} \\ 0 & 0 & k_{31} & k_{32} & k_{33} & k_{34} \\ 0 & 0 & k_{41} & k_{42} & k_{43} & k_{44} \end{matrix} \begin{bmatrix} v_1 \\ \phi_1 \\ v_2 \\ \phi_2 \\ v_3 \\ \phi_3 \end{bmatrix} = [A]_2 [U]_2$$

$S_1 \quad S_2 \quad S_3 \quad S_4$

Единой формулой составление матрицы  $[A]$  можно записать в виде  $A_{S_i, S_j} = k_{ij}$ , где  $S_i = S(i)$  - элементы списка индексов.

Но в программах для ЭВМ массив  $[A]$  не создается отдельно для каждого

элемента, а создается общая матрица  $[A]$  и ее элементы вычисляются по формуле  $A_{s,s_j}^n = A_{s,s_j}^{n-1} + k_{ij}$ , где  $n$ - номер текущего цикла, т.е. идет накопление. Всего число циклов равно числу элементов.

Запишем сумму 1-го и 2-го столбца

$$\begin{bmatrix} Q_1 \\ M_1 \\ Q_2 \\ M_2 \\ 0 \\ 0 \end{bmatrix} + \begin{bmatrix} 0 \\ 0 \\ Q_1 \\ M_1 \\ Q_2 \\ M_2 \end{bmatrix} = \begin{bmatrix} k_{11}^1 & k_{12}^1 & k_{13}^1 & k_{14}^1 & 0 & 0 \\ k_{21}^1 & k_{22}^1 & k_{23}^1 & k_{24}^1 & 0 & 0 \\ k_{31}^1 & k_{32}^1 & k_{33}^1 + k_{11}^2 & k_{34}^1 + k_{12}^2 & k_{13}^2 & k_{14}^2 \\ k_{41}^1 & k_{42}^1 & k_{43}^1 + k_{21}^2 & k_{44}^1 + k_{22}^2 & k_{23}^2 & k_{24}^2 \\ 0 & 0 & k_{31}^2 & k_{32}^2 & k_{33}^2 & k_{43}^2 \\ 0 & 0 & k_{41}^2 & k_{42}^2 & k_{43}^2 & k_{44}^2 \end{bmatrix} \begin{bmatrix} v_1 \\ \phi_1 \\ v_2 \\ \phi_2 \\ v_3 \\ \phi_3 \end{bmatrix} = [A][U] \Leftrightarrow [A][U] = [B]$$

Учитывая, что  $v_1 = 0, \phi_1 = 0, v_3 = 0, \phi_3 = 0$

$$\begin{bmatrix} Q_1 \\ M_1 \\ Q_2 \\ M_2 \\ 0 \\ 0 \end{bmatrix} + \begin{bmatrix} 0 \\ 0 \\ Q_1 \\ M_1 \\ Q_2 \\ M_2 \end{bmatrix} = \begin{bmatrix} 0 & 0 & k_{13}^1 & k_{14}^1 & 0 & 0 \\ 0 & 0 & k_{23}^1 & k_{24}^1 & 0 & 0 \\ 0 & 0 & k_{33}^1 + k_{11}^2 & k_{34}^1 + k_{12}^2 & 0 & 0 \\ 0 & 0 & k_{43}^1 + k_{21}^2 & k_{44}^1 + k_{22}^2 & 0 & 0 \\ 0 & 0 & k_{31}^2 & k_{32}^2 & 0 & 0 \\ 0 & 0 & k_{41}^2 & k_{42}^2 & 0 & 0 \end{bmatrix} \begin{bmatrix} 0 \\ 0 \\ v_2 \\ \phi_2 \\ 0 \\ 0 \end{bmatrix} = [A][U] \Leftrightarrow [A][U] = [B]$$

Для 3 столбца

$$\begin{bmatrix} Q_R \\ M_R \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \begin{bmatrix} Q_R \\ M_R \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix};$$

Для 4 столбца

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ Q_R \\ M_R \end{bmatrix} = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ Q_R \\ M_R \end{bmatrix}$$

Подставляя выражения для 1,2,3 и 4 столбцов, в уравнения равновесия и проводя суммирование, получим

$$\begin{bmatrix} 1 & 0 & k_{13}^1 & k_{14}^1 & 0 & 0 \\ 0 & 1 & k_{23}^1 & k_{24}^1 & 0 & 0 \\ 0 & 0 & k_{33}^1 + k_{11}^2 & k_{34}^1 + k_{12}^2 & 0 & 0 \\ 0 & 0 & k_{43}^1 + k_{21}^2 & k_{44}^1 + k_{22}^2 & 0 & 0 \\ 0 & 0 & k_{31}^2 & k_{32}^2 & 1 & 0 \\ 0 & 0 & k_{41}^2 & k_{42}^2 & 0 & 1 \end{bmatrix} \begin{bmatrix} Q_R^A \\ M_R^A \\ v_2 \\ \phi_2 \\ Q_R^B \\ Q_R^B \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ P \\ 0 \\ 0 \\ 0 \end{bmatrix} \Leftrightarrow [A][U] = [B]$$

Просмотрев структуру системы уравнений можно увидеть, что для определения неизвестных перемещений  $v_2, \phi_2$ , нужно взять 3 и 4 уравнения.

$$\begin{bmatrix} k_{33}^1 + k_{11}^2 & k_{34}^1 + k_{12}^2 \\ k_{43}^1 + k_{21}^2 & k_{44}^1 + k_{22}^2 \end{bmatrix} \begin{bmatrix} v_2 \\ \phi_2 \end{bmatrix} = \begin{bmatrix} P \\ 0 \end{bmatrix} \Leftrightarrow [A][U] = [B]$$

Следовательно, систему уравнений можно составлять только для неизвестных перемещений. Для этого нумеровать нужно только неизвестные перемещения. В этом случае список индексов имеет вид

№ элемента	Список индексов			
	S(1)	S(2)	S(3)	S(4)
1	0	0	1	2

2 1 2 0 0

## Разработка программы расчета.

В программах на ЭВМ построение матрицы  $[A]$  записывается следующим образом

```
10 DIM K(4,4)      массив для матрицы [K]
20 DIM S(4)        массив для списка индексов [S]
30 READ N, NE      ввод числа уравнений и элементов
35 DIM A(N,N)      массив для матрицы [A]
40 FOR J = 1 TO NE  цикл по элементам
50   READ S(1),S(2),S(3),S(4)  ввод списка индексов матрицы
                               (адреса матрицы [K] в
                               матрице [A])
60   READ DL, EI   ввод длины и жесткости
70   T = EI / DL^3  коэффициент при матрице [K]
                               матрица жесткости
80   K(1,1)=12: K(1,2)=6*DL: K(1,3)=-12: K(1,4)=6*DL
90   K(2,1)=6*DL: K(2,2)=4*DL^2: K(2,3)=-6*DL: K(2,4)=2*DL^2
100  K(3,1)=-12: K(3,2)=-6*DL: K(3,3)=12: K(3,4)=-6*DL
110  K(4,1)=6*DL: K(4,2)=2*DL^2: K(4,3)=-6*DL: K(4,4)=4*DL^2
                               сборка матрицы [A]
120  FOR L = 1 TO 4
130   FOR M = 1 TO 4
140    A(S(L), S(M)) = A(S(L), S(M)) + K(L, M) * T
150   NEXT M
160  NEXT L
170 NEXT J
```

## Ввод нагрузок

Для ввода нагрузок нужно составить матрицу нагрузок  $[B]$  и ввести ее в ЭВМ Программа для ЭВМ для ввода матрицы нагрузок имеет вид

```
180 DIM B(N)      массив для матрицы нагрузок
190 FOR J = 1 TO N
200  READ B(J)     ввод матрицы нагрузок
220 NEXT J
```

## Решение системы линейных уравнений

Решение системы уравнений можно выполнить по программе.

```
230 DIM U(N)      массив для неизвестных перемещений
240 'ПРЯМОЙ ХОД
250 FOR J = 1 TO N - 1
260  FOR JJ = J+1 TO N
270   XM = A(J, JJ) / A(J, J)
280   FOR I = JJ TO N
290    A(JJ, I) = A(JJ, I) - A(J, I) * XM
300  NEXT I
310  B(JJ) = B(JJ) - B(J) * XM
320 NEXT JJ
330 NEXT J
340 'ОБРАТНЫЙ ХОД
350 U(N) = B(N) / A(N, N)
360 FOR J = N - 1 TO 1 STEP -1
```

```

370 SUM = 0
380 FOR I = J + 1 TO N
390 SUM = SUM + A(J, I) * U(I)
400 NEXT I
410 U(J) = (B(J) - SUM) / A(J, J)
420 NEXT J

```

### Построение эпюр внутренних усилий

Эпюра моментов строится для каждого элемента отдельно по формуле  $M_x = Q_1x + M_1$ , где  $Q_1, M_1$  находится из 1 и 2 уравнения соотношения (2). Эпюра поперечных сил по формуле  $Q_x = Q_1$ .

```

500 RESTORE 690                начать чтение данных с 690
                                строки
510 FOR J = 1 TO NE            цикл по элементам
520 PRINT "Element "; J       печать номера элемента
530 READ S(1), S(2), S(3), S(4) ввод списка индексов матрицы
540 READ DL, EI                ввод длины и жесткости
550 V1=U(S(1))                Выборка из общей матрицы переме-
560 V2=U(S(2))                щений, перемещений относящихся
570 V3=U(S(3))                к данному элементу
580 V4=U(S(4))
590 FOR X = 0 TO DL STEP (DL / 6) Эпюры строятся в 7 точках
600 Q1=EI/DL^3 * ( 12*V1 + 6*DL*V2 - 12*V3 + 6*DL*V4)
610 M1=EI/DL^3 * (-6*DL*V1 - 4*DL^2*V2 + 6*DL*V3 - 2*DL^2*V4)
620 M = Q1*X + M1
630 Q = Q1
640 PRINT "x=";X, "M=";M, "Q=";Q вывод результатов
650 NEXT X
660 NEXT J
670 END                        конец программы

```

### Исходные данные к примеру.

```

680 DATA 2,2                  число уравнений, число элементов
690 DATA 0,0,1,2              список индексов для 1-го элемента
700 DATA 3,1                  длина и жесткость EI для 1-го
                                элемента
710 DATA 1,2,0,0              список индексов для 2-го элемента
720 DATA 3,1                  длина и жесткость EI для 2-го
                                элемента
730 DATA 1,0                  элементы матрицы нагрузок

```

Для проведения расчета надо собрать текст программы, убрать боковые комментарии и выполнить ее используя программу "gwbasic.exe" или другую программу использующую язык BASIC.


### Результаты расчета.

Element	1		
x= 0	M= .75	Q=-.5	
x= .5	M= .5	Q=-.5	
x= 1	M= .25	Q=-.5	
x= 1.5	M= 0	Q=-.5	
x= 2	M=-.25	Q=-.5	
x= 2.5	M=-.5	Q=-.5	
x= 3	M=-.75	Q=-.5	

Element 2

x= 0	M=-.75	Q= .5
x= .5	M=-.5	Q= .5
x= 1	M=-.25	Q= .5
x= 1.5	M= 0	Q= .5
x= 2	M= .25	Q= .5
x= 2.5	M= .5	Q= .5
x= 3	M= .75	Q= .5

Ok\_



Приложение 2